

Le caractère & définit la lettre suivante comme raccourci dans le menu déroulant lors de la navigation via le clavier.

## Programmation VBA

# Tips 2

### Exemple de réalisation d'une macro avec SolidWorks

#### Fonctionnalité de la macro :

Cet exemple de macro permet de calculer le nombre de faces et le nombre de facettes d'une pièce.

On peut gérer le nombre de facettes d'une pièce en diminuant ou en augmentant la résolution d'image ombré et la qualité d'ébauche LCS/LCA. On trouve cette commande dans le menu « Outil\Options » et ensuite dans « Propriétés du document » et « Qualité d'image ».

#### Réalisation de la macro :

Pour réaliser une macro dans SolidWorks, on peut grâce à la barre d'outils « Macro » lancer l'éditeur de macro qui va ouvrir Microsoft Visual Basic.




## 1) Ajout des références :

1. Dans le menu **Outils**, cliquez sur **Référence** pour afficher la boîte de dialogue **Référence**
2. On ajoute :
  - SolidWorks 2005 Constant type library
  - SolidWorks 2005 type library
  - SolidWorks 2005 exposed type libraries for add-in use

## 2) Réalisation de l'interface utilisateur :

Cette fenêtre va permettre à l'utilisateur de créer l'interface de sa macro. Une boîte à outils permet d'ajouter différents contrôles à cette fenêtre. Grâce à celle-ci on va pouvoir rédiger le code.



## 3) Réalisation du code :

La touche « Ok » va générer le programme donc c'est là que le code va être rédigé. Pour cela on réalise un double clic sur la touche « OK » et cela ouvre la fenêtre de code.

Dans les « Option Explicit » on déclare les variables générales du code qui vont être utilisées dans différentes procédures.

The image shows a VBA code editor window titled 'nombre\_face - frmMain (Code)'. It has two tabs: '(Général)' and '(Déclarations)'. The '(Déclarations)' tab is active, showing the following code:

```
Option Explicit
Dim intFace      As Integer
Dim intPiece     As Integer
Dim lgFacette    As Long
```

## Code :

Dans « cmdOk\_Click » on déclare et on affecte les variables « swApp » et « swDoc » qui vont permettre la liaison avec SolidWorks, ensuite on rédige son code et on déclare au fur et à mesure les variables dont on a besoin pour stocker les données.

```

Private Sub cmdOk_Click()
    Dim swApp      As SolidWorks.SldWorks      'déclaration des variables
    Dim swDoc      As SolidWorks.ModelDoc2
    Dim swRootComponent As SolidWorks.Component2
    Dim swConfig   As SolidWorks.Configuration
    Dim swAssy     As SolidWorks.AssemblyDoc
    Dim swBody     As SolidWorks.body2

    'affectation des variables
    Set swApp = Application.SldWorks
    Set swDoc = swApp.ActiveDoc

    'on vide les zones de texte
    intFace = 0
    intPiece = 0
    lgFacette = 0

    'si il n'y a pas de document ouvert alors on sort de la fonction
    If swDoc Is Nothing Then
        MsgBox "il n'y a pas de document ouvert"
        Exit Sub
    End If

    Select Case swDoc.GetType
        'si le document est une pièce alors y va calculer le nombre de face(swDoc)
        Case swDocPART
            'on va à la fonction "corps"
            Call corps(swDoc)

            'si le document est un assemblage alors y va à traverseComposant
            Case swDocASSEMBLY
                Set swConfig = swDoc.GetActiveConfiguration
                Set swRootComponent = swConfig.GetRootComponent
                'on va à la fonction "alléger"
                Call alléger(swAssy)
                'on va à la fonction "traverseComposant"
                Call traverseComposant(swRootComponent)
                'on rend visible le texte et le label de "nombre de pièce"
                txtTexte1.Visible = True
                Label1.Visible = True
                'on met le nombre de pièce trouvé dans txtTexte1
                txtTexte1.Text = intPiece

            'si le document est une mise en plan alors msgbox et on sort
            Case swDocDRAWING

                MsgBox " cette macro ne s'exécute que sur une pièce ou un assemblage"

        End Select

        'on met le nombre de face et de facette trouvé dans leur zone de texte
        txtTexte0.Text = intFace
        txtFacette.Text = lgFacette

    End Sub

```

On teste si il y a un document d'ouvert avec la fonction « If ». Si il n'y en a pas on sort du programme.

A l'aide de la fonction « Select Case » on identifie le type de document SolidWorks ouvert.

On affiche le résultat.

## Le document ouvert est une pièce :

```

Sub corps(ByVal swDoc As SolidWorks.ModelDoc2)

    Dim swBody           As SolidWorks.body2
    Dim swBodiesCorps    As Variant
    Dim swBodiesSurf     As Variant
    Dim intNbr           As Integer
    Dim vrInfos          As Variant
    Dim a                As Integer
    Dim swFonction       As SolidWorks.feature

    ' swDoc.Visible = True
    'on sélectionne les corps volumique et on va à nombreDeFace
    swBodiesCorps = swDoc.GetBodies2(swBodyType_e.swSolidBody, False)
    If IsEmpty(swBodiesCorps) = False Then
        intNbr = UBound(swBodiesCorps)

        For a = 0 To intNbr
            Set swBody = swBodiesCorps(a)
            nombreDeFace swBody
        Next a
    End If

    'on sélectionne les corps surfacique et on va à nombreDeFace
    swBodiesSurf = swDoc.GetBodies2(swSheetBody, False)
    If IsEmpty(swBodiesSurf) = False Then
        intNbr = UBound(swBodiesSurf)
        For a = 0 To intNbr
            Set swBody = swBodiesSurf(a)
            nombreDeFace swBody
        Next a
    End If

End Sub

```

On sélectionne dans l'arbre de création chaque corps et chaque surface de la pièce et on va à « nombreDeFace » pour calculer le nombre de faces.

On crée une boucle « For-Next » pour sélectionner les corps et pour chaque corps on va à « nombreDeFace ».

On réalise ensuite la même boucle pour sélectionner les surfaces et pour chaque surface on va à « nombreDeFace ».

Une fois que la fonction est terminée on retourne soit à « cmdOk\_click » si le document de base était une pièce soit à « traverseComponent » si le document était un assemblage.

## On calcule le nombre de face et facette :

```

Sub nombreDeFace(ByVal swBody As SolidWorks.body2)

    Dim vrInfo          As Variant
    Dim swFace          As SolidWorks.face2
    Dim intNbr          As Integer
    Dim intNombre       As Integer
    Dim svTessellation  As SolidWorks.Tessellation
    Dim vrRetVal        As Variant
    Dim swFaceId        As Object
    Dim retVal          As Long

    Set swFace = swBody.GetFirstFace
    'on boucle pour connaître le nombre de face du corps ou de la surface
    Do
        'compteur pour connaître le nombre de face
        intNombre = intNombre + 1
        'ou
        vrRetVal = swFace.GetTriangleCount()
        'compteur qui ajoute pour chaque pièce le nombre de facette
        lgFacette = lgFacette + vrRetVal
        'on boucle sur la face suivante
        Set swFace = swFace.GetNextFace
        'on sort dès qu'il n'y a plus de face
    Loop Until swFace Is Nothing
    'compteur qui additionne le nombre de face pour 1 corps ou 1 surface,
    'avec le nombre total de face
    intFace = intFace + intNombre

End Sub

```

Pour trouver le nombre de face sur le corps ou la surface alors on utilise la fonction « GetFirstFace » pour sélectionner la première face, ensuite on réalise une boucle de type « Do... Loop » et la fonction « GetNextFace » pour boucler sur toutes les faces.

Boucle qui permet de passer sur toutes les faces du corps.

Les variables « lgFacette » et « intFace » ont été définies dans les « Option explicit » pour pouvoir être récupérées dans « cmdOk » et être affichées dans la fenêtre. Lorsque « nombreDeFace » est terminé on retourne à « corps ».

### Le document ouvert est un assemblage:

Lorsque le document est un assemblage alors on va à « alléger » pour régler l'assemblage sur résolu. Ensuite on va à « traverseComposant » pour lister chaque pièce de l'assemblage et des sous assemblages et pour chacune d'elles on exécute le même code que lorsque le document d'origine est une pièce. C'est-à-dire que l'on va à « corps ».

```

Sub alléger(ByVal swAssy As SldWorks.AssemblyDoc)

    Dim swApp          As SldWorks.SldWorks
    Dim swModel        As SldWorks.ModelDoc2
    Dim nRetVal        As Long

    Set swApp = Application.SldWorks
    Set swModel = swApp.ActiveDoc
    Set swAssy = swModel

    'on règle les pièces de l'assemblage sur résolu
    nRetVal = swAssy.ResolveAllLightWeightComponents(False)

End Sub

Sub traverseComposant(ByVal swComposant As SldWorks.Component2)

    Dim vrChild        As Variant
    Dim swDoc           As SldWorks.ModelDoc2
    Dim a              As Integer
    Dim intNombre1     As Integer

    'dans l'assemblage on boucle pour récupérer chaque pièces et chaque sous assemblage...
    If swComposant.Name <> "" Then
        'on regarde si c'est une pièce ou un sous-assemblage
        Set swDoc = swComposant.GetModelDoc
        If swDoc.GetType = swDocPART Then
            intNombre1 = intNombre1 + 1
            'on appelle corps
            Call corps(swDoc)
        End If
    End If

    vrChild = swComposant.GetChildren
    'on calcule le nombre de pièce dans l'assemblage
    intPiece = intPiece + intNombre1

    'si il y a un sous assemblage on retourne à traverseComposant
    For a = 0 To UBound(vrChild)
        intPiece = intPiece + intNombre1
        ' on boucle
        Call traverseComposant(vrChild(a))
    Next a
End Sub
    
```

On règle les pièces de l'assemblage sur résolu.

Si le composant de l'assemblage est une pièce alors on va à corps.

Si le composant de l'assemblage est un sous assemblage alors on boucle les pièces du sous assemblage en appelant « traverseComposant ».


Une fois que la fonction est terminée on retourne à « cmdOk\_click » pour afficher les valeurs calculées.

### Le document ouvert est une mise en plan:

On affiche alors un message pour informer l'utilisateur que cette macro ne fonctionne que sur une pièce ou un assemblage.

### 4) On lance le programme :

On lance le programme avec « Exécuter Sub/UserForm ».



Pour tester le programme on peut placer des points d'arrêt en face du code comme on le voit ci-dessus, ils permettent au programme de s'arrêter sur ces points.

On peut ensuite exécuter le programme pas à pas en appuyant sur la touche F8.